# Maintainer:
# Contributor: Bartłomiej Piotrowski <bpiotrowski@archlinux.org>

pkgname=rpcsvc-proto
pkgver=1.4.4
pkgrel=1
pkgdesc='rpcsvc protocol definitions from glibc'
arch=(loong64 x86_64)
url='https://github.com/thkukuk/rpcsvc-proto'
license=(BSD)
depends=(glibc)
source=(https://github.com/thkukuk/rpcsvc-proto/archive/v$pkgver/$pkgname-$pkgver.tar.gz)
sha512sums=('8477709678a5af6338011e730a20ba5fc5e97f6f360738a956fd96ad26da309135ff042c65b3590bc35fee907e2eea7203aa782963efd30d066aa1306c2c5326')

prepare() {
  cd $pkgname-$pkgver
  autoreconf -fi
}

build() {
  cd $pkgname-$pkgver
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname-$pkgver
  make DESTDIR="$pkgdir" install
  install -Dm644 COPYING "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
